Laboratório 1

Preparação do trabalho

- O GNU Linear Programming Kit (GLPK) já é instalado em Windows e Linux.
- Para Windows é necessário modificar o ambiente (`set PATH=%PATH%;<dir>`).
  Para a versão 4.45: `<dir>=c:"\Program Files\glpk-4.45\w64"`.
- Teste: Executar `glpsol --help`.
- A documentação sobre os formatos CPLEX lp e GNU MathProg (sub-linguagem de AMPL) é disponível em http://www.inf.ufrgs.br/~mrpritt/oc/lang.pdf e no apêndice das notas de aula.

Exemplo

Para demonstrar a especificação de um problema nos formatas CPLEX lp e GNU mathprog, considere o exemplo de um importador de Whisky:

Um importador de Whisky tem as seguintes restrições de importação

- no máximo 2000 garrafas de *Johnny Ballantine* por 70 R$ cada uma,
- no máximo 2500 garrafas de *Old Gargantua* por 50 R$ cada uma,
- no máximo 1200 garrafas de *Misty Deluxe* por 40 R$ cada uma.

Dos Whiskies importados ele produz três misturas A, B, C, que ele vende por 68 R$, 57 R$ e 45 R$, respectivamente. As misturas são

- A: no mínimo 60% Johnny Ballantine, no máximo 20% Misty Deluxe,
- B: no mínimo 15% Johnny Ballantine, no máximo 60% Misty Deluxe,
- C: no máximo 50% Misty Deluxe.

Quais seriam as misturas ótimas, e quantas garrafas de cada mistura devem ser produzidas para maximizar o lucro?

A formulação do problema é

$$
\begin{array}{lll}
\textbf{max} & 68x_A + 57x_B + 45x_C - (70x_J + 50x_O + 40x_M) & \\
\textbf{s.a} & x_J \leq 2000 & \\
 & x_O \leq 2500 & \\
 & x_M \leq 1200 & \\
 & x_{J,A} \geq 0.6x_A & \\
 & x_{M,A} \leq 0.2x_A & \\
 & x_{J,B} \geq 0.15x_B & \\
 & x_{M,B} \leq 0.6x_B & \\
 & x_{M,C} \leq 0.5x_C & \\
 & x_m = x_{m,A} + x_{m,B} + x_{m,C} & m \in \{J, O, M\} \\
 & x_m = x_{J,m} + x_{O,m} + x_{M,m} & m \in \{A, B, C\} \\
 & x_{m,n} \geq 0 & m \in \{J, O, M\}, n \in \{A, B, C\}
\end{array}
$$

Especificação com CPLEX lp

http://www.inf.ufrgs.br/~mrpritt/oc/whisky.lp

```
Maximize
        lucro:  68xa+57xb+45xc-70xj-50xo-40xm
Subject to
        limiteJohnny:   xj<=2000
        limiteOldGarg:  xo<=2500
        limiteMisty:    xm<=1200

        percJohnnyA:    xja-0.6xa>=0
        percMistzA:     xma-0.2xa<=0
        percJohnnyB:    xjb-0.15xb>=0
        percMistyB:     xmb-0.6xb<=0
        percMistyC:     xmc-0.5xc<=0

        totalJohnny:    xj-xja-xjb-xjc=0
        totalOldGarg:   xo-xoa-xob-xoc=0
        totalMisty:     xm-xma-xmb-xmc=0
        totalA:         xa-xja-xoa-xma=0
        totalB:         xb-xjb-xob-xmb=0
        totalC:         xc-xjc-xoc-xmc=0
Bounds
        0 <= xja
        0 <= xjb
        0 <= xjc
        0 <= xma
        0 <= xmb
        0 <= xmc
        0 <= xoa
        0 <= xob
        0 <= xoc
End
```

Especificação com GNU mathprog

http://www.inf.ufrgs.br/~mrpritt/oc/whisky.mod

```
# conjuntos basicos de indices
set misturas;
set marcas   ;

# custo de cada garafa por marca
param custoMarca { marcas } >= 0;
# preco de cada mistura
param precoMistura { misturas } >= 0;
# limite de importacao de cada marca
param limiteMarca { marcas } >= 0;
# percentagens minimas das marcas nas misturas
param percMin { marcas, misturas } >= 0, <= 1, default 0;
# percentages maximas das marcas nas misturas
param percMax { marcas, misturas } >= 0, <= 1, default 1;

# numero de garrafas
var numMistura { misturas } >= 0;
var numMarca   { marcas } >= 0;
```

```
var numGarrafas { marcas, misturas } >= 0;

# definicao do modelo
maximize lucro:
        sum { j in misturas } precoMistura[j]*numMistura[j] -
        sum { j in marcas } custoMarca[j]*numMarca[j];

subject to limitesImportacao { m in marcas }:
        numMarca[m] <= limiteMarca[m];
subject to maxPercentagens { m in marcas, i in misturas }:
        numGarrafas[m,i] <= percMax[m,i]*numMistura[i];
subject to minPercentagens { m in marcas, i in misturas }:
        numGarrafas[m,i] >= percMin[m,i]*numMistura[i];
subject to totalMarca { m in marcas }:
        numMarca[m] = sum { j in misturas } numGarrafas[m,j];
subject to totalMistura { m in misturas }:
        numMistura[m] = sum { j in marcas } numGarrafas[j,m];

# definicao dos dados
data;

# formato de tabela: a 1a coluna define o conjunto marcas
# e os seguintes colunas os valores de parametros que
# dependem de marcas
param : marcas : custoMarca limiteMarca :=
        'J'     70          2000
        'O'     50          2500
        'M'     40          1200    ;

# tabela para misturas
param : misturas : precoMistura :=
        'A'         68
        'B'         57
        'C'         45     ;

# definicao de percMin em formato de matriz:
# a 1a linha contem os 1os indices da matriz
# a 1a coluna contem os 2os indices da matriz
# o valor "." significa usar o default
param percMin : 'A'  'B'   'C' :=
        'J'     0.6  0.15   .
        'O'     .    .      .
        'M'     .    .      .   ;

# matrix das percentagens maximos
param percMax : 'A'  'B'  'C' :=
        'J'     .    .    .
        'O'     .    .    .
        'M'     0.20 0.60 0.50  ;

end;
```

Questão 1 (Empresa de aço)
Uma empresa de aço produz placas e canos de ferro. As taxas de produção semanal são 200t/h para placas e 140t/h para canos. O lucro desses produtos e 25\$/t para placas e 30\$/t para canos. Considerando a demanda atual, os limites de produção semanal são 6000t de placas e 4000t de canos. A jornada de produção semanal é de 40h. Quantas toneladas de placas e canos devem ser produzidas para maximizar o lucro?

- Coloque o problema em formato CPLEX lp (p.ex. arquivo `e1.lp`) e resolve com

```
glpsol --cpxlp e1.lp -o e1.sol
```

- Coloque o problema em formato MathProg (p.ex. arquivo `e1.mod`) e resolve com

```
glpsol -m e1.mod -o e1.sol
```

Questão 2 (Refinar óleo (da Costa))
Um certo óleo é refinado a partir da mistura de outros óleos, vegetais ou não vegetais. Temos óleos vegetais V1 e V2 e óleos não vegetais NV1 NV2 NV3. Por restrições da fábrica, um máximo de 200 toneladas de óleos vegetais podem ser refinados por mês, e um máximo de 250 toneladas de óleos não vegetais. A acidez do óleo desejado deve estar entre 3 e 6 (dada uma unidade de medida) e a acidez depende linearmente das quantidades/acidez dos óleos brutos usados. O preço de venda de uma tonelada do óleo é R\$ 150. Calcule a mistura que maximiza o lucro, dado que:

Óleo	V1	V2	NV1	NV2	NV3
Custo/ton	110	120	130	110	115
Acidez	8.8	6.1	2.0	4.2	5.0

Formule e resolve este problema.

- Coloque o problema em formato CPLEX lp (p.ex. arquivo `ro.lp`) e resolve com

```
glpsol --cpxlp ro.lp -o ro.sol
```

Questão 3
Um estudante, na véspera de seus exames finais, dispõe de 100 horas de estudo para dedicar às disciplinas A, B e C. Cada um destes exames é formado por 100 questões, e o estudante espera acertar, alternativamente, uma questão em A, duas em B ou três em C, por cada hora de estudo. Suas notas nas provas anteriores foram 6, 7 e 10, respectivamente, e sua aprovação depende de atingir uma média mínima de 5 pontos em cada disciplina. O aluno deseja distribuir seu tempo de forma a ser aprovado com a maior soma total de notas.

- Coloque o problema em formato MathProg e resolve.

Questão 4 (McDonalds)
Resolve o problema da dieta: Suponha que temos uma tabela de nutrientes de diferentes tipos de alimento. Sabendo o valor diário de referência (VDR) de cada nutriente e o preço de cada unidade de alimento, qual a dieta ótima, i.e. que contém ao menos o valor diário de referência, mas custo de menos? Com m nutrientes e n alimentos, seja a_{ij} a quantidade de nutriente i na alimento j, r_i valor diário de referência do nutriente i e c_j o preço de alimento j. Queremos saber as quantidades x_j de cada alimento que

$$\textbf{minimiza} \qquad c_1x_1 + \cdots + c_1x_n$$

sujeito a

$$\begin{aligned} a_{11}x_1 + \cdots + a_{1n}x_n &\geq r_1 \\ &\cdots \\ a_{m1}x_1 + \cdots + a_{mn}x_n &\geq r_m \\ x_1, \ldots, x_n &\geq 0 \end{aligned}$$

Os dados do McDonalds são disponíveis em http://www.inf.ufrgs.br/~mrpritt/oc/McDonalds.dat

no formato AMPL. Os dados usam os nomes `cost[i]` para c_i, `amt[i,j]` para a_{ij} e `n_min[i]` para r_i. O valores dos nutrientes são em % do VDR, por isso podemos usar simplesmente $r_i = 100\%$. Além disso, o arquivo contém os dados `f_min[i]`, `f_max[i]` e `n_max[i]` para o valor mínimo e máximo de alimento da cada tipo, e a percentagem máxima de cada nutriente. Essas colunas contém um "`.`" que significa, que o valor "default" é usado. Esse valor tem que ser especificado no modelo.

- Implemente o problema da dieta usando AMPL (p.ex. arquivo `McDonalds.mod`).
- Resolve o problema com `glpsol -m McDonalds.mod -d McDonalds.dat -o McDonalds.sol` usando GLPK.
- Qual a dieta ótima com as restrições acima? Qual o seu custo?

Para resolver os seguintes itens, modifique o sistema adequadamente e resolve de novo.

- O que custa uma dieta ótima entre 2000 kcal (o VDR) e 2100 kcal?
- Qual a dieta ótima vegetariana?
- Qual a dieta ótima em que cada alimento é consumido uma única vez?

Questão 5 (Companhia de torneiras)
A Companhia de Torneiras Bica Larga fabrica uma linha completa de metais sanitários. A companhia tem duas fábricas, uma no sul e uma no nordeste. Designe pelo índice i a Fábrica i. O vice-presidente industrial Sacy Assed, quer planejar a programação de produção de uma torneira particular para os próximos dois trimestres, que são designados como Períodos 1 e 2. Em cada período, a capacidade de produção na Fábrica i é L_i.
A companhia envia suas torneiras a dois armazéns distribuidores, designados pelo índice j. As exigências mínimas de demanda a serem atingidas no fim do Período t no Armazém j é R_{tj} onde $t = 1$ e 2, e $j = 1$ e 2.
Para cada fábrica i, faça

x_i quantidade produzida *e embarcada* no fim do primeiro trimestre

y_i quantidade produzida *e embarcada* no fim do segundo trimestre

s_i quantidade produzida *e armazenada* na Fábrica i no fim do primeiro trimestre

Para cada Fábrica i e Armazém j, faça

z_{ij} quantidade embarcada da Fábrica i ao Armazém j no fim do primeiro trimestre

v_{ij} quantidade embarcada da Fábrica i ao Armazém j no fim do segundo trimestre

Para cada Armazém j, faça

w_j quantidade armazenada no Armazém j no fim do primeiro trimestre

Define quaisquer outros símbolos que você precisar para representar as variáveis do problema.
Sacy exige que a soma total de produção para a Fábrica 1 nos dois trimestres combinados seja pelo menos 85% da quantidade correspondente para a Fábrica 2. Em virtude do espaço limitado de armazenagem, não mais que S_i torneiras podem ser armazenadas na Fábrica i no fim do primeiro trimestre.
O custo de produção de uma torneira na Fábrica i é p_i. O custo de uma torneira armazenada no fim do primeiro trimestre na Armazém j é h_j. Embarcar um item da Fábrica i ao Armazém j implica um custo de transporte c_{ij} por torneira. Suponha que o tempo de transporte seja virtualmente zero (comparado ao comprimento do trimestre), de modo que um item embarcado durante um trimestre chega antes do trimestre termine.

- Mostre como uma programação ótima de produção e embarque pode ser achada de uma formulação de programação linear.
- Coloque o problema em formato AMPL (p.ex. arquivo `ea.mod`). Usa na formulação os nomes dos parâmetros definidos na descrição acima (`L`,`R`,`S`,`p`,`h`,`c`)
- Define alguns valores dos dados mencionados (p.ex. arquivo `ea.dat`)
- Resolve o problema com

```
glpsol -m ea.mod -d ea.dat -o ea.sol
```